# $Id$
# Maintainer: Andrea Scarpino <andrea@archlinux.org>

pkgname=libmusicbrainz5
pkgver=5.1.0
pkgrel=3
pkgdesc="Library That Provides Access to the MusicBrainz Server"
arch=('x86_64')
url='https://musicbrainz.org/doc/libmusicbrainz'
license=('LGPL2.1')
depends=('neon' 'libxml2')
makedepends=('cmake')
source=("https://github.com/metabrainz/libmusicbrainz/releases/download/release-${pkgver}/libmusicbrainz-$pkgver.tar.gz")
sha256sums=('6749259e89bbb273f3f5ad7acdffb7c47a2cf8fcaeab4c4695484cef5f4c6b46')

prepare() {
  mkdir build
}

build() {
  cd build
  cmake ../libmusicbrainz-${pkgver} \
    -DCMAKE_INSTALL_PREFIX=/usr \
    -DCMAKE_BUILD_TYPE=Release
  make
}

package() {
  cd build
  make DESTDIR="$pkgdir" install
}
